
O Comando dsolve
O comando básico do Maple para resolver equações diferenciais ordinarias eh o "

dsolve ". 

A sintaxe do dsolve é

dsolve("oque", "como");

O "oque" refere-se a EDO (ou sistema de edo's) junto com condições iniciais. O 
"como" especifica qual rotina do Maple vai ser utilizada.

É útil dar nomes a todas as equações e condições iniciais quando usamos o dsolve . 
Exemplos:

> eq:=diff(y(x),x)=x*y(x);

> init:=y(2)=1;

Agora, eq é o nome da edo que queremos resolver e init é o nome das condições 
iniciais. É

importante usar y(x) -- isto indica que tomamos y como uma variavel dependente e x
como variavel independente. 
Para resolver sem a condição inicial, isto é achar a solução geral, usamos `dsolve` 

> dsolve(eq,y(x));

Note que _C1 é a constante arbitraria produzida pelo maple. 

Para resolver o PVI devemos agrupar a EDO junto com a condicao inicial entre chaves.



> dsolve({eq,init},y(x));

As vezes Maple dá uma resposta implicita, assim é comum pedir que o Maple explicite 
a resposta em função da variavel dependente. Exemplo:

> dsolve(diff(y(x),x)=y(x)^2, y(x));

Usamos o comando explicit como opcao em dsolve :

> dsolve(diff(y(x),x)=y(x)^2, y(x), explicit);

Assim é melhor!!!

Uso mais avançado de dsolve:

Existem 4 maneiras de usar o dsolve para situações além das edos de primeira ordem e 
pvi simples.

1. Equações de ordem superior 

2. Sistemas de EDO's

3. Soluções numericas

4. Usando séries de potências para obter soluções de edo's

1. Equações de ordem superior : edo's de ordem 2 ou superior podem ser resolvidas 
usando o dsolve . As derivadas de ordem superior são escritas como no exemplo.

> diff(f(x),x$3);



Uma EDO de ordem 2:

> eqn2:=diff(y(x),x$2)+3*diff(y(x),x)+2*y(x)=exp(x);

A solução geral é:

> dsolve(eqn2,y(x));

Para este problema vamos especificar condições de y em x=1:

> inits:=y(1)=2, D(y)(1)=4;

> dsolve({eqn2,inits},y(x));

2. Sistemas de EDO's : Vamos ver um exemplo.

> eqns:=diff(y(x),x)+diff(z(x),x)=x, diff(y(x),x)-2*diff(z(x),x)=x^2;

> inits:= y(0)=1, z(0)=2;

> dsolve({eqns,inits},{y(x),z(x)});

3. Soluções Numéricas : Em geral é impossivel obter explicitamente a soluçao de uma 
EDO, neste caso usamos um método numérico para aproximar a solução. O Maple faz 
isto, basta usar a opcao numeric em dsolve , como no exemplo:

> eqn:=diff(y(x),x)+exp(y(x))*x^3=2*sin(x); init:=y(0)=2;



> F:=dsolve({eqn,init},y(x),numeric);

O valor de F em x=2 é

> F(2);

Em geral é util plotar numericamente a solução obtida de uma EDO. Para fazer 
isto usamos o comando odeplot ao resultado (F neste caso) de dsolve(....,numeric) .
Para usar

odeplot precisamos de "chamar" o plots .

> with(plots,odeplot);

Para plotar a solucao usamos a sintaxe.

> odeplot(F,[x,y(x)],-2..2);

4 . Usando series de potencias. Exemplo: resolver y'+x*y=0, y(0)=1 por series .



> dsolve({diff(y(x),x)+x*y(x)=0, y(0)=1},y(x), series);

Se voce quer mais termos, peça mais.

> Order:=14: dsolve({diff(y(x),x)+x*y(x)=0, y(0)=1},y(x), series);

> 


